# How does this TransformedDistribution compare to the Normal distribution?

I am working with the following TransformedDistribution:

Dist = TransformedDistribution[2 v S1 + 2 v S2 - v c, {S1 \[Distributed] BinomialDistribution[1/2 (c + t), p], S2 \[Distributed] BinomialDistribution[1/2 (c - t), 1 - p]}]


This random variable has neat expressions for mean, variance, skewness, and kurtosis:

(-1 + 2 p) t v

-4 c (-1 + p) p v^2

((1 - 2 p) t v)/(c Sqrt[-c (-1 + p) p v^2])

3 - 6/c + 1/(c p - c p^2)


What I am trying to do is compare this random variable, when $$t=0$$, $$c = \frac{1}{v^2}$$, $$v$$ approaches $$0$$, and $$c$$ thus approaches infinity, with one drawn from the Normal distribution. Under these restrictions, $$\mu$$ = 0, $$\sigma^2 = 4 (1 - p) p$$, $$\lambda = 3$$, and $$\kappa = 0$$. Thus, it appears we are dealing with the Normal distribution, but that is not necessarily the case (see:link).

• The distribution is never exactly normal. So what distance metric would characterize an important difference and what size of difference would be important? I ask because "how close is close" is a subject matter issue rather than a statistical issue. – JimB Dec 29 '18 at 22:49
• @JimB That is a good question. Suppose I wanted to just inspect visually. For example, if I also set p = 0.5 then variance becomes 1 and a comparison to the Standard Normal distribution would seem appropriate. Could a comparison be setup using Manipulate? – user120911 Dec 29 '18 at 22:53
• That's the "I'll know it when I see it" approach. It's used all the time. (But it's not necessarily a consistent approach.) My point is that there's no universally accepted rule. – JimB Dec 29 '18 at 22:58
• The answer from @user120911 below shows that there is an equivalence in distribution in the limit. Do you also need to know "how close" you are for small values of $v$? – JimB Dec 30 '18 at 18:54
• @JimB That is interesting, but unless you already have that worked out, I would not request it. – user120911 Dec 30 '18 at 19:21

Dist = FullSimplify[TransformedDistribution[2 v S1 + 2 v S2 - v c, {S1 \[Distributed] BinomialDistribution[1/2 (c + t), p], S2 \[Distributed] BinomialDistribution[1/2 (c - t), 1 - p]}]]


The moment generating function of the Dist is

MomentGeneratingFunction[Dist, x]

E^(-c v x) (-E^(2 v x) (-1 + p) + p)^((c - t)/2) (1 + (-1 + E^(2 v x)) p)^((c + t)/2)


Now let us consider the situation where c = 1/v^2, v-> 0, and t = 0. Under these restrictions, the moment generating function of the Dist is

Limit[With[{c = 1/v^2, t = 0},E^(-c v x) (-E^(2 v x) (-1 + p) + p)^((c - t)/2) (1 + (-1 + E^(2 v x)) p)^((c + t)/2)], v -> 0]


which simplifies to

E^(-2 (-1 + p) p x^2)


In comparison, if we examine the Normal Distribution with variance given by -4(-1 + p)p (i.e., the variance of Dist under the stated restrictions), then we notice the moment generating function is

MomentGeneratingFunction[NormalDistribution[0, Sqrt[-4 (-1 + p) p ]],x]

E^(-2 (-1 + p) p x^2)


which we notice is equal to the above.

• Did you obtain the mgf from Mathematica commands? If so, would you consider adding those to your answer? – JimB Dec 30 '18 at 18:50
• Excellent! Thank you. – JimB Dec 30 '18 at 20:41